Illustration of proof that AB and BA have same characteristic polynomial except for a power of the variable, so that the nonzero eigenvalues are the same. We start by loading the LinearAlgebra package and inializing some useful matrices.

```
> with(LinearAlgebra):
> x3:=Matrix(1..3,1..3,shape=scalar[x]);
```

$$x3 := \begin{bmatrix} x & 0 & 0 \\ 0 & x & 0 \\ 0 & 0 & x \end{bmatrix}$$

```
> x4:=Matrix(1..4,1..4,shape=scalar[x]);
```

$$x4 := \begin{bmatrix} x & 0 & 0 & 0 \\ 0 & x & 0 & 0 \\ 0 & 0 & x & 0 \\ 0 & 0 & 0 & x \end{bmatrix}$$

Now, define a couple of random matrices.

```
> A:=<<3,1,-2>|<2,0,5>|<7,-1,3>|<4,1,-5>>;
```

$$A := \begin{bmatrix} 3 & 2 & 7 & 4 \\ 1 & 0 & -1 & 1 \\ -2 & 5 & 3 & -5 \end{bmatrix}$$

```
> B:=<<3,0,9,7>|<2,-1,-3,6>|<7,3,-2,4>>;
```

$$B := \begin{bmatrix} 3 & 2 & 7 \\ 0 & -1 & 3 \\ 9 & -3 & -2 \\ 7 & 6 & 4 \end{bmatrix}$$

This is the basic matrix used in the proof that is built from A and B.

```
> MAB:=<<x3|A>,<B|x4>>;
```

$$MAB := \begin{bmatrix} x & 0 & 0 & 3 & 2 & 7 & 4 \\ 0 & x & 0 & 1 & 0 & -1 & 1 \\ 0 & 0 & x & -2 & 5 & 3 & -5 \\ 3 & 2 & 7 & x & 0 & 0 & 0 \\ 0 & -1 & 3 & 0 & x & 0 & 0 \\ 9 & -3 & -2 & 0 & 0 & x & 0 \\ 7 & 6 & 4 & 0 & 0 & 0 & x \end{bmatrix}$$

```
> Determinant(MAB);
```

$$-36875\,x - 652\,x^3 - 86\,x^5 + x^7$$

This determinant is a single expression that will be related to the characteristic polynomials of AB and BA.

More names for matrices needed in the construction: zero matices and identity matrices of appropriate sizes.

```
> B0:=Matrix(4,3,0);
```

$$B0 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> A0:=Transpose(B0);
```

$$A0 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> I3:=IdentityMatrix(3);
```

$$I3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> I4:=IdentityMatrix(4);
```

$$I4 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

This is one of the left multipliers.

```
> LB:=<<I3|A0>,<-B|x4>>;
```

$$LB := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ -3 & -2 & -7 & x & 0 & 0 & 0 \\ 0 & 1 & -3 & 0 & x & 0 & 0 \\ -9 & 3 & 2 & 0 & 0 & x & 0 \\ -7 & -6 & -4 & 0 & 0 & 0 & x \end{bmatrix}$$

... and the product with the basis matrix.

```
> LB.MAB;
```

$$\begin{bmatrix} x & 0 & 0 & 3 & 2 & 7 & 4 \\ 0 & x & 0 & 1 & 0 & -1 & 1 \\ 0 & 0 & x & -2 & 5 & 3 & -5 \\ 0 & 0 & 0 & 3+x^2 & -41 & -40 & 21 \\ 0 & 0 & 0 & 7 & -15+x^2 & -10 & 16 \\ 0 & 0 & 0 & -28 & -8 & -60+x^2 & -43 \\ 0 & 0 & 0 & -19 & -34 & -55 & -14+x^2 \end{bmatrix}$$

The determinant wil have three factors of x from the first three rows and the determinant of the lower right 4 by 4 block. That block is x^2 times the identity plus a numerical matrix. As expected by considering the block multiplication, that numerical matrix is -BA.

```
> -B.A;
```

$$\begin{bmatrix} 3 & -41 & -40 & 21 \\ 7 & -15 & -10 & 16 \\ -28 & -8 & -60 & -43 \\ -19 & -34 & -55 & -14 \end{bmatrix}$$

```
> Determinant(%%);
```

$$-36875\,x^5 - 652\,x^7 - 86\,x^9 + x^{11}$$

This confirms that the determinant is the product of the determinants of the two matrix factors. On the other hand, it should be x^3 times the result of substituting x^2 for the variable in the cahracteristic polynomial for BA.

```
> expand(x^3*subs(y=x^2,CharacteristicPolynomial(B.A,y)));
```

$$-36875\,x^5 - 652\,x^7 - 86\,x^9 + x^{11}$$

Now, consider the other left muliplier.

```
> LA:=<<x3|-A>,<B0|I4>>;
```

$$LA := \begin{bmatrix} x & 0 & 0 & -3 & -2 & -7 & -4 \\ 0 & x & 0 & -1 & 0 & 1 & -1 \\ 0 & 0 & x & 2 & -5 & -3 & 5 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> LA.MAB;
```

$$\begin{bmatrix} x^2-100 & -7 & -29 & 0 & 0 & 0 & 0 \\ -1 & -11+x^2 & -13 & 0 & 0 & 0 & 0 \\ 14 & 48 & 25+x^2 & 0 & 0 & 0 & 0 \\ 3 & 2 & 7 & x & 0 & 0 & 0 \\ 0 & -1 & 3 & 0 & x & 0 & 0 \\ 9 & -3 & -2 & 0 & 0 & x & 0 \\ 7 & 6 & 4 & 0 & 0 & 0 & x \end{bmatrix}$$

```
> -A.B;
```

$$\begin{bmatrix} -100 & -7 & -29 \\ -1 & -11 & -13 \\ 14 & 48 & 25 \end{bmatrix}$$

This time, we have the following expression that should be equal to the determinant.

```
> expand(x^4*subs(y=x^2,CharacteristicPolynomial(A.B,y)));
```

$$-36875\,x^4 - 652\,x^6 - 86\,x^8 + x^{10}$$

We see this is x^3 times the determinant of MAB, as expected. We made some substitutions to get these expressions, so it might be helpful to see the actual characteristic polynomials.

```
> CharacteristicPolynomial(A.B,y);
```

$$-36875 - 652\,y - 86\,y^2 + y^3$$

```
> CharacteristicPolynomial(B.A,y);
```

$$-36875\,y - 652\,y^2 - 86\,y^3 + y^4$$

```
>
```